

## Référencement des objets-application

POURQUOI ? COMMENT ?

LIAISON ANTICIPEE ou TARDIVE ?


Nous savons que l'un des atouts de la suite **Microsoft Office** réside dans l'**interopérabilité** des applications qui la compose.


Lorsque nous désirons **programmer en VBA** l'interaction d'une application MS Office avec une autre, par exemple d'EXCEL avec WORD, il est nécessaire de référencer depuis EXCEL **les éléments** (propriétés, méthodes, collections, énumérations, ...) **exposées par l'objet** WORD.

Ce référencement est nécessaire quelque soient les applications **Microsoft Office** en jeu mais également lorsque nous utilisons des **objets Microsoft** autres (« FileSystemObject », « Microsoft XML » ...) ou encore lorsque nous mettons en œuvre des **composants non Microsoft** (composants ActiveX par exemple).

Cette fiche « **Bonnes pratiques VBA** » présente de façon concise ce qu'il est nécessaire de connaître sur le référencement des objets.

*(Les copies-écrans du présent document sont issues d'EXCEL version 365)*

L'icône  marque les paragraphes plus particulièrement dédiés à une **utilisation professionnelle d'EXCEL**. Si tel n'est pas le cas, ils peuvent être ignorés sans inconvénient.



L'icône  signale une remarque réclamant une **attention particulière**.

## Table des matières

LA BOÎTE DE DIALOGUE « REFERENCES » .....	3
LA LIAISON ANTICIPEE (EARLY BINDING) .....	4
LA LIAISON TARDIVE (LATE BINDING) .....	6
LIAISON ANTICIPEE vs TARDIVE – AVANTAGES/INCONVENIENTS .....	7
INTELLISENSE c'est quoi ? .....	7
L'EXPLORATEUR D'OBJETS c'est quoi ? .....	8
UN MOT SUR LES PERFORMANCES .....	9
PROBLEMES LIES A LA VERSION MS OFFICE .....	9
LIAISON ANTICIPEE vs TARDIVE – LE BON COMPROMIS .....	10

## LA BOÎTE DE DIALOGUE « REFERENCES »

Commençons par ouvrir la **boîte de dialogue « Références »** depuis le menu de l'interface VBE (Visual Basic Editor) : >Outils/Références...



Toutes les bibliothèques d'objets des **applications installées et enregistrées** sur notre ordinateur apparaissent dans la liste déroulante de la boîte de dialogue « Références – VBAProject ».

Remarquons que quelques-unes d'entre-elles sont **sélectionnées automatiquement** et, parmi celles-ci, la référence « **Microsoft Excel XX.X Object Library** » : celle de l'application EXCEL (XX.X étant la version d'EXCEL installée sur notre ordinateur. Ici la version 16.0 - vous pouvez en avoir une différente).


Si nous en avons l'utilité dans notre projet, nous pouvons sélectionner n'importe laquelle des bibliothèques d'objets présentes en la **cochant sa case** dans cette liste.

En réalisant cette opération, nous mettons en œuvre une **liaison anticipée** avec cet objet-application.

### LA LIAISON ANTICIPEE (EARLY BINDING)

A titre d'exemple, mettons en œuvre une **liaison anticipée avec les objets de l'application WORD**.

Recherchons la bibliothèque d'objets dans la liste déroulante et sélectionnons-la :



(Ici la WORD est en version 16.0 - vous pouvez en avoir une différente)

Cliquons sur **OK**.

## Tutoriel « EXCELLONS » : Le référencement des objets-application

---

Dans le squelette de code suivant, nous mettons en œuvre **la liaison anticipée** aux objets exposés par WORD :

```
Sub gvs_Exemple_Referencement_Liaison_Anticipée()  
  'On déclare une variable objet référençant l'application  
  Dim oWordApp As Word.Application  
  'On déclare une variable objet référençant un document word  
  Dim oDoc As Word.Document  
  
  'On affecte une nouvelle instance de l'application WORD  
  Set oWordApp = New Word.Application  
  'On crée un nouveau document que l'on affecte à la variable  
  Set oDoc = oWordApp.Documents.Add  
  '  
  '  
  ' Partie de notre programme consommant les objets exposés  
  '  
  '  
  oDoc.Close False ' A adapter  
  'On ferme l'instance de l'application utilisée  
  oWordApp.Quit  
  
  'On fait le ménage  
  Set oDoc = Nothing  
  Set oWordApp = Nothing  
End Sub
```

### LA LIAISON TARDIVE (LATE BINDING)

Avec le mode liaison tardive, **nous n'intervenons pas au préalable** dans la boîte de dialogue « Références ».









Pour reprendre l'exemple avec l'application WORD, dans un **module du projet VBA**, le code en **mode liaison tardive** se présente de la façon suivante :

```
Sub gvs_Exemple_Referencement_Liaison_Tardive()
    'On déclare une variable objet pour l'application
    Dim oWordApp As Object
    'On déclare une variable objet référençant un document word
    Dim oDoc As Object
    'On déclare un booleen
    Dim booWORDActif As Boolean

    'On suspend la survenue d'erreur
    On Error Resume Next
    'On se réfère à l'application WORD en considérant qu'elle est déjà ouverte
    booWORDActif = True
    Set oWordApp = GetObject("Word.Application")
    'Si la variable oWordapp n'a pas été initialisée par la ligne précédente c'est que WORD n'est pas ouvert.
    If oWordApp Is Nothing Then
        'On affecte alors une nouvelle instance de l'application WORD
        Set oWordApp = CreateObject("Word.Application")
        booWORDActif = False
    End If
    'On rétablit la survenue d'erreur
    On Error GoTo 0
    'On affecte un nouveau document
    Set oDoc = oWordApp.Documents.Add
    '
    ' Partie de notre programme consommant les objets exposés
    '
    oDoc.Close False
    'On ferme l'instance de l'application utilisée si WORD n'était pas préalablement actif
    If Not booWORDActif Then
        oWordApp.Quit
    End If

    'On fait le ménage
    Set oDoc = Nothing
    Set oWordApp = Nothing
End Sub
```

## LIAISON ANTICIPEE vs TARDIVE – AVANTAGES/INCONVENIENTS

	VBE IntelliSense	Objets disponibles dans l'explorateur d'objets	Performances	Dépendance à la version MS OFFICE
LIAISON ANTICIPEE				
LIAISON TARDIVE				

## INTELLISENSE c'est quoi ?


Avec le mode liaison anticipée, nous **bénéfisons des facilités offertes par « IntelliSense »**.

IntelliSense est le nom d'un ensemble de fonctionnalités conçues pour nous faciliter le codage.

Après avoir frappé un « . », l'éditeur **VBE propose automatiquement** tous les éléments de l'objet référencé puis, les sélectionne en fonction des lettre que nous frappons :


```
'On affecte une nouvelle instance de l'application WORD
Set oWordApp = New Word.Application
'On affecte une nouveau document
Set oDoc = oWordApp.Documents.

'
' Notre programme agissant
'
oDoc.Close False
'On ferme l'instance de l'ap
oWordApp.Quit
```




## L'EXPLORATEUR D'OBJETS c'est quoi ?


Nous accédons à l'Explorateur d'objet à partir du menu VBE : >Affichage/Explorateur d'objets (ou bien F2) :



Nous obtenons alors la fenêtre suivante, dans laquelle nous pouvons sélectionner un objet particulier (ici WORD) :



On obtient alors l'intégralité des éléments exposés par l'objet (ici WORD) :





## UN MOT SUR LES PERFORMANCES

Même si **cet aspect ne constitue pas un critère déterminant** dans le choix de la méthode de liaison à utiliser, il peut être noté que, par rapport aux classeurs « liaison tardive », les classeurs « liaison anticipée »:

- Devraient **s'exécuter légèrement plus rapidement**
- Auront une taille **légèrement plus volumineuse**.

## PROBLEMES LIES A LA VERSION MS OFFICE

Microsoft respecte (la plupart du temps) le principe de **compatibilité ascendante**, mais ce principe **ne s'applique pas dans le sens descendant**.


Si, nous développons une solution avec **liaisons anticipées** sous **MS Office 2015**, par exemple, celle-ci fonctionnera sans problème sur des ordinateurs installés avec **MS Office 2015 et ultérieurs** (2016, 365...). Lorsque la version **MS Office est postérieure**, **MS Office modifiera lui-même** le référencement initial au profit d'un référencement à la version installée.

En revanche, sur les ordinateurs installés avec **MS Office 2013** ou version plus ancienne, nos utilisateurs risquent fortement de nous revenir avec : « ...ça ne marche pas ton truc...J'ai le message d'erreur 'zigloub (☹) n'est pas une fonction valide' ...». Lorsque la version **MS Office est antérieure**, nos références seront marquées par MS Office comme « **MANQUANT** ». Pour corriger, nous devons intervenir sur chaque ordinateur pour **déréférencer** les références initiales et **référencer la version MS Office installée**. (Si le nombre d'ordinateurs est important, c'est galère ! ☹)

### LIAISON ANTICIPEE vs TARDIVE – LE BON COMPROMIS

Comme constaté plus, nous avons intérêt, **dans la plupart de nos projets**, à opter pour les **liaisons anticipées**. Toutefois, lorsque nous développons une **solution destinée à être distribuée**, nous **privilegierons les liaisons tardives** et nous procéderons de la façon suivante :

- Lors de la **phase de développement** et jusqu'à la finalisation de nos tests, pour bénéficier des avantages liés à l'écriture du code, nous mettrons en œuvre **les liaisons anticipées**.
- Avant de distribuer notre solution, nous transformerons notre code pour mettre en œuvre **les liaisons tardives** et **déréférencerons** le (les) objet(s) référencé(s) en le(s) décochant dans la liste des références :



**i** Après cette opération, ayons soin de **RECOMPILER la solution** !

**Attention aux énumérations** utilisées : **après déréférencement**, à la compilation, elles généreront une erreur « variable non définie ». Il faudra les **définir explicitement** en tant que constantes.